

[The division algorithm for $\mathbb{Q}[x]$ in Maple

```
> DIV := proc(a::polynom(rational,x),b::polynom(rational,x))
  local q,r,t,lt;
  if b=0 then error "division by zero" fi;
  lt := proc(a) lcoeff(a,x)*x^degree(a,x) end;
  r := a;
  q := 0;
  while r <> 0 and degree(r,x) >= degree(b,x) do
    t := lt(r) / lt(b);
    r := r - expand(t*b);
    q := q + t;
  od;
  return(r,q);
end;
```

```
> DIV( 10*x^2+2*x+1, 5*x-3 );
```

$$\frac{29}{5}, 2x + \frac{8}{5}$$

```
> degree(0,x);
```

$-\infty$

[The natural Euclidean algorithm for $\mathbb{Q}[x]$ in Maple.

```
> EA := proc(a::polynom(rational,x),b::polynom(rational,x))
  local q,r,k,monic;
  monic := proc(a) if a=0 then 0 else a/lcoeff(a,x) fi; end;
  r[0] := a;
  r[1] := b;
  k := 1;
  while r[k] <> 0 do
    (r[k+1],q) := DIV(r[k-1],r[k]);
    k := k+1; print(r[k]);
  od;
  monic(r[k-1]);
end;
```

```
> a := expand((13*x^4-7*x^3+12*x+11)*(x-2));
```

```
b := expand((5*x^3-22*x^2+31*x+5)*(x-2));
```

$$a := 13x^5 - 33x^4 + 14x^3 + 12x^2 - 13x - 22$$

$$b := 5x^4 - 32x^3 + 75x^2 - 57x - 10$$

```
> EA(a,b);
```

$$\frac{3507}{25}x^3 - \frac{2964}{5}x^2 + \frac{14632}{25}x + \frac{392}{5}$$

$$\frac{33619625}{4099683}x^2 - \frac{59194675}{4099683}x - \frac{109450}{27889}$$

$$\frac{1968734291223553x}{45211167405625} - \frac{3937468582447106}{45211167405625}$$

0

$x-2$

[>